
Solution to Exercise 2.9.3

We denote the irreducible characters of G := HS by χ1, . . . , χ24, ordered as
in the ATLAS.

gap> ct := CharacterTable("HS");;

gap> PermChars(ct,100) = [Sum( Irr(ct){[1,2,3]} )];

true

gap> ratchars := Irr(ct){[1,2,3]};;

gap> ratchars := Set( Tensored( ratchars, ratchars) );;

gap> ratchars := Set( Tensored( ratchars, ratchars) );;

gap> mult1 := Union( List( MatScalarProducts( ct, Irr(ct), ratchars ) ,

> y -> Positions( y, 1) ));

[ 1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12, 13, 16, 17, 20, 23 ]

χ1, χ2, χ3, being the irreducible constituents of the multiplicity free permu-
tation character of degree 100, are characters of rational representations. Of
course products of characters of rational representations are also characters of
rational representations. Thus ratchars is a set of characters of rational rep-
resentations.

We have computed the set

mult1 := { i ∈ {1, . . . , 24} | (χi, ψ)G = 1 for some ψ ∈ ratchars }.

From Theorem 2.9.19 we know that mQ(χi) = 1 for i ∈ mult1. In case χi is
rational valued, it is then by Lemma 2.9.15 a character of a rational representa-
tion. So we obtain a larger supply of characters of a rational representations.:

gap> ratirr := Filtered( mult1, i -> Irr(ct)[i] in RationalizedMat(Irr(ct)) );

[ 1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 13, 16, 17, 20, 23 ]

gap> ratchars := Set( Tensored ( Irr(ct){ratirr}, Irr(ct){ratirr} ) );;

gap> Union( List( MatScalarProducts( ct, Irr(ct), ratchars ) ,

> y -> Positions( y, 1) )) = [1..24] ;

true

Thus every χi ∈ Irr(G) occurs with multiplicity one in a character of a rational
representation and so has Schur index one.

